1: $\int \frac{A + B \text{Log}\left[c (d + e x)^n\right]}{\sqrt{a + b \text{Log}\left[c (d + e x)^n\right]}} \, dx$

Rule:

$$\int \frac{A + B \text{Log}\left[c (d + e x)^n\right]}{\sqrt{a + b \text{Log}\left[c (d + e x)^n\right]}} \, dx \longrightarrow$$

$$\frac{B (d + e x) \sqrt{a + b \text{Log}\left[c (d + e x)^n\right]}}{b e} + \frac{2 A b - B (2 a + b n)}{2 b} \int \frac{1}{\sqrt{a + b \text{Log}\left[c (d + e x)^n\right]}} \, dx$$

–

Program code:

```
Int[(A_.+B_.*Log[c_.*(d_.+e_.*x_)^n_.])/Sqrt[a_+b_.*Log[c_.*(d_.+e_.*x_)^n_.]],x_Symbol] :=
  B*(d+e*x)*Sqrt[a+b*Log[c*(d+e*x)^n]]/(b*e) +
   (2*A*b-B*(2*a+b*n))/(2*b)*Int[1/Sqrt[a+b*Log[c*(d+e*x)^n]],x] /;
FreeQ[{a,b,c,d,e,A,B,n},x]
```

Rules for integrands of the form $u\ (a+b\,\mathrm{Log}[c\,x^n])^p$

4. $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$

0: $\int x^m\left(d+\frac{e}{x}\right)^q(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $m==q\land q\in\mathbb{Z}$

- Derivation: Algebraic simplification

- Rule: If $m==q\land q\in\mathbb{Z}$, then

$$\int x^m\left(d+\frac{e}{x}\right)^q(a+b\,\mathrm{Log}[c\,x^n])^p\,dx\longrightarrow\int(e+d\,x)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$$

- Program code:

```
Int[x_^m_.*(d_+e_./x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[(e+d*x)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[m,q] && IntegerQ[q]
```

1: $\int x^m (d + e x^r)^q (a + b \operatorname{Log}[c x^n]) \, dx$ when $q \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{Log}[c x^n]) == \frac{b n}{x}$

Rule: If $q \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, let $u \to \int x^m (d + e x^r)^q \, dx$, then

$$\int x^m (d + e x^r)^q (a + b \operatorname{Log}[c x^n]) \, dx \longrightarrow u (a + b \operatorname{Log}[c x^n]) - b n \int \frac{u}{x} \, dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^r_.)^q_.*Log[c_.*x_^n_.],x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x^r)^q,x]},
  Dist[Log[c*x^n],u,x] - n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{c,d,e,n,r},x] && IGtQ[q,0] && IntegerQ[m] && Not[EqQ[q,1] && EqQ[m,-1]]
```

```
Int[x_^m_.*(d_+e_.*x_^r_.)^q_.*(a_+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x^r)^q,x]},
  u*(a+b*Log[c*x^n]) - b*n*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[q,0] && IntegerQ[m] && Not[EqQ[q,1] && EqQ[m,-1]]
```

2: $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx$ when $m+r\,(q+1)+1=0\land m\neq -1$

Derivation: Integration by parts

Basis: If $m+r\,(q+1)+1=0\land m\neq -1$, then $(f\,x)^m \left(d+e\,x^r\right)^q = \partial_x \frac{(f\,x)^{m+1} \left(d+e\,x^r\right)^{q+1}}{d\,f\,(m+1)}$

Rule: If $m+r\,(q+1)+1=0\land m\neq -1$, then

$$\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx \longrightarrow \frac{(f\,x)^{m+1} \left(d+e\,x^r\right)^{q+1} \left(a+b\,\text{Log}\left[c\,x^n\right]\right)}{d\,f\,(m+1)} - \frac{b\,n}{d\,(m+1)} \int (f\,x)^m \left(d+e\,x^r\right)^{q+1} dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])/(d*f*(m+1)) -
  b*n/(d*(m+1))*Int[(f*x)^m*(d+e*x^r)^(q+1),x] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m+r*(q+1)+1,0] && NeQ[m,-1]
```

3. $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$ when $m == r-1 \land p \in \mathbb{Z}^+$

   1. $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$ when $m == r-1 \land p \in \mathbb{Z}^+ \land (m \in \mathbb{Z} \lor f > 0)$

      1: $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$ when $m == r-1 \land p \in \mathbb{Z}^+ \land (m \in \mathbb{Z} \lor f > 0) \land r == n$

Derivation: Integration by substitution

Rule: If $m == r-1 \land p \in \mathbb{Z}^+ \land (m \in \mathbb{Z} \lor f > 0) \land r == n$, then

$$\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx \longrightarrow \frac{f^m}{n}\,\text{Subst}\left[\int (d+e\,x)^q\,(a+b\,\text{Log}[c\,x])^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*x_^n_])^p_.,x_Symbol] :=
  f^m/n*Subst[Int[(d+e*x)^q*(a+b*Log[c*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m,r-1] && IGtQ[p,0] && (IntegerQ[m] || GtQ[f,0]) && EqQ[r,n]
```

2. $\int (f x)^m (d + e x^r)^q (a + b \operatorname{Log}[c x^n])^p \, dx$ when $m == r - 1 \land p \in \mathbb{Z}^+ \land (m \in \mathbb{Z} \lor f > 0) \land r \neq n$

1: $\int \frac{(f x)^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx$ when $m == r - 1 \land p \in \mathbb{Z}^+ \land (m \in \mathbb{Z} \lor f > 0) \land r \neq n$

Derivation: Integration by parts

Basis: $\frac{(f x)^m}{d+e x^r} == \frac{f^m}{e r} \partial_x \operatorname{Log}\left[1 + \frac{e x^r}{d}\right]$

Rule: If $m == r - 1 \land p \in \mathbb{Z}^+ \land (m \in \mathbb{Z} \lor f > 0) \land r \neq n$, then

$$\int \frac{(f x)^m (a + b \operatorname{Log}[c x^n])^p}{d + e x^r} \, dx \longrightarrow \frac{f^m \operatorname{Log}\left[1 + \frac{e x^r}{d}\right] (a + b \operatorname{Log}[c x^n])^p}{e r} - \frac{b f^m n p}{e r} \int \frac{\operatorname{Log}\left[1 + \frac{e x^r}{d}\right] (a + b \operatorname{Log}[c x^n])^{p-1}}{x} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_./(d_+e_.*x_^r_),x_Symbol] :=
  f^m*Log[1+e*x^r/d]*(a+b*Log[c*x^n])^p/(e*r) -
  b*f^m*n*p/(e*r)*Int[Log[1+e*x^r/d]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,f,m,n,r},x] && EqQ[m,r-1] && IGtQ[p,0] && (IntegerQ[m] || GtQ[f,0]) && NeQ[r,n]
```

2: $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$ when $m == r-1\ \land\ p \in \mathbb{Z}^+ \land\ (m \in \mathbb{Z}\ \lor\ f>0)\ \land\ r \neq n\ \land\ q \neq -1$

- Derivation: Integration by parts

- Rule: If $m == r-1\ \land\ p \in \mathbb{Z}^+ \land\ (m \in \mathbb{Z}\ \lor\ f>0)\ \land\ r \neq n\ \land\ q \neq -1$, then

$$\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx \ \longrightarrow\ \frac{f^m \left(d+e\,x^r\right)^{q+1} \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{e\,r\,(q+1)} - \frac{b\,f^m\,n\,p}{e\,r\,(q+1)} \int \frac{\left(d+e\,x^r\right)^{q+1} \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p-1}}{x}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  f^m*(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])^p/(e*r*(q+1)) -
  b*f^m*n*p/(e*r*(q+1))*Int[(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m,r-1] && IGtQ[p,0] && (IntegerQ[m] || GtQ[f,0]) && NeQ[r,n] && NeQ[q,-1]
```

2: $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$ when $m == r-1\ \land\ p \in \mathbb{Z}^+ \land\ \neg\ (m \in \mathbb{Z}\ \lor\ f>0)$

Derivation: Piecewise constant extraction

- Rule: If $m == r-1\ \land\ p \in \mathbb{Z}^+ \land\ \neg\ (m \in \mathbb{Z}\ \lor\ f>0)$, then

$$\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx \ \longrightarrow\ \frac{(f\,x)^m}{x^m} \int x^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$$

- Program code:

```
Int[(f_*x_)^m_.*(d_+e_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (f*x)^m/x^m*Int[x^m*(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && EqQ[m,r-1] && IGtQ[p,0] && Not[(IntegerQ[m] || GtQ[f,0])]
```

4. $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $q+1\in\mathbb{Z}^-$

1: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\mathrm{Log}[c\,x^n])\,dx$ when $q+1\in\mathbb{Z}^-\wedge m>0$

Rule: If $q+1\in\mathbb{Z}^-\wedge m>0$, then

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\mathrm{Log}[c\,x^n])\,dx\,\longrightarrow$$
$$\frac{(f\,x)^m\,(d+e\,x)^{q+1}\,(a+b\,\mathrm{Log}[c\,x^n])}{e\,(q+1)}-\frac{f}{e\,(q+1)}\int (f\,x)^{m-1}\,(d+e\,x)^{q+1}\,(a\,m+b\,n+b\,m\,\mathrm{Log}[c\,x^n])\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (f*x)^m*(d+e*x)^(q+1)*(a+b*Log[c*x^n])/(e*(q+1)) -
  f/(e*(q+1))*Int[(f*x)^(m-1)*(d+e*x)^(q+1)*(a*m+b*n+b*m*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && ILtQ[q,-1] && GtQ[m,0]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx$ when $q+1\in\mathbb{Z}^- \land m\in\mathbb{Z}^-$

Rule: If $q+1\in\mathbb{Z}^- \land m\in\mathbb{Z}^-$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx \longrightarrow$$

$$-\frac{(f\,x)^{m+1}\left(d+e\,x^2\right)^{q+1}\left(a+b\,\text{Log}\left[c\,x^n\right]\right)}{2\,d\,f\,(q+1)}+\frac{1}{2\,d\,(q+1)}\int (f\,x)^m \left(d+e\,x^2\right)^{q+1}\left(a\,(m+2\,q+3)+b\,n+b\,(m+2\,q+3)\,\text{Log}\left[c\,x^n\right]\right) dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(q+1)*(a+b*Log[c*x^n])/(2*d*f*(q+1)) +
  1/(2*d*(q+1))*Int[(f*x)^m*(d+e*x^2)^(q+1)*(a*(m+2*q+3)+b*n+b*(m+2*q+3)*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && ILtQ[q,-1] && ILtQ[m,0]
```

5: $\int x^m \left(d+e x^2\right)^q \left(a+b \operatorname{Log}\left[c x^n\right]\right) dx$ when $\frac{m}{2} \in \mathbb{Z} \land q-\frac{1}{2} \in \mathbb{Z} \land \neg (m+2q<-2 \lor d>0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(d+e x^2\right)^q}{\left(1+\frac{e}{d} x^2\right)^q} == 0$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land q-\frac{1}{2} \in \mathbb{Z} \land \neg (m+2q<-2 \lor d>0)$, then

$$\int x^m \left(d+e x^2\right)^q \left(a+b \operatorname{Log}\left[c x^n\right]\right) dx \to \frac{d^{\text{IntPart}[q]} \left(d+e x^2\right)^{\text{FracPart}[q]}}{\left(1+\frac{e}{d} x^2\right)^{\text{FracPart}[q]}} \int x^m \left(1+\frac{e}{d} x^2\right)^q \left(a+b \operatorname{Log}\left[c x^n\right]\right) dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  d^IntPart[q]*(d+e*x^2)^FracPart[q]/(1+e/d*x^2)^FracPart[q]*Int[x^m*(1+e/d*x^2)^q*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[m/2] && IntegerQ[q-1/2] && Not[LtQ[m+2*q,-2] || GtQ[d,0]]
```

```
Int[x_^m_.*(d1_+e1_.*x_)^q_*(d2_+e2_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (d1+e1*x)^q*(d2+e2*x)^q/(1+e1*e2/(d1*d2)*x^2)^q*Int[x^m*(1+e1*e2/(d1*d2)*x^2)^q*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[d2*e1+d1*e2,0] && IntegerQ[m] && IntegerQ[q-1/2]
```

6. $\int \frac{(d+e x^r)^q \, (a+b \operatorname{Log}[c x^n])^p}{x} \, dx$ when $p \in \mathbb{Z}^+$

1. $\int \frac{(a+b \operatorname{Log}[c x^n])^p}{x \, (d+e x^r)} \, dx$ when $p \in \mathbb{Z}^+$

1: $\int \frac{a+b \operatorname{Log}[c x^n]}{x \, (d+e x^r)} \, dx$ when $\frac{r}{n} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\frac{F[x^n]}{x} == \frac{1}{n} \operatorname{Subst}\left[\frac{F[x]}{x}, x, x^n\right] \partial_x x^n$

Rule: If $\frac{r}{n} \in \mathbb{Z}$, then

$$\int \frac{a+b \operatorname{Log}[c x^n]}{x \, (d+e x^r)} \, dx \longrightarrow \frac{1}{n} \operatorname{Subst}\left[\int \frac{a+b \operatorname{Log}[c x]}{x \, (d+e x^{r/n})} \, dx, x, x^n\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_])/(x_*(d_+e_.*x_^r_.)),x_Symbol] :=
  1/n*Subst[Int[(a+b*Log[c*x])/(x*(d+e*x^(r/n))),x],x,x^n] /;
FreeQ[{a,b,c,d,e,n,r},x] && IntegerQ[r/n]
```

2: $\int \frac{(a+b \operatorname{Log}[c x^n])^p}{x \, (d+e x)} \, dx$ when $p \in \mathbb{Z}^+$

Rule: Algebraic expansion

Basis: $\frac{1}{x \, (d+e x)} == \frac{1}{d x} - \frac{e}{d \, (d+e x)}$

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{\left(a+b \operatorname{Log}\left[c x^n\right]\right)^p}{x (d+e x)} \, dx \longrightarrow \frac{1}{d} \int \frac{\left(a+b \operatorname{Log}\left[c x^n\right]\right)^p}{x} \, dx - \frac{e}{d} \int \frac{\left(a+b \operatorname{Log}\left[c x^n\right]\right)^p}{d+e x} \, dx$$

–

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(x_*(d_+e_.*x_)),x_Symbol] :=
  1/d*Int[(a+b*Log[c*x^n])^p/x,x] - e/d*Int[(a+b*Log[c*x^n])^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0]
```

x: $\int \frac{(a + b \operatorname{Log}[c x^n])^p}{x (d + e x^r)} dx$ when $p \in \mathbb{Z}^+$

– Rule: Integration by parts

■ Basis: $\frac{1}{x (d+e x^r)} == \partial_x \frac{r \operatorname{Log}[x] - \operatorname{Log}\left[1+\frac{e x^r}{d}\right]}{d r}$

– Basis: $\partial_x (a + b \operatorname{Log}[c x^n])^p == \frac{b n p (a+b \operatorname{Log}[c x^n])^{p-1}}{x}$

– Note: This rule returns antiderivatives in terms of x^r instead of x^{-r}, but requires more steps and larger antiderivatives.

– Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \operatorname{Log}[c x^n])^p}{x (d + e x^r)} dx \longrightarrow$$

$$\frac{\left(r \operatorname{Log}[x] - \operatorname{Log}\left[1 + \frac{e x^r}{d}\right]\right) (a + b \operatorname{Log}[c x^n])^p}{d r} - \frac{b n p}{d} \int \frac{\operatorname{Log}[x] (a + b \operatorname{Log}[c x^n])^{p-1}}{x} dx + \frac{b n p}{d r} \int \frac{\operatorname{Log}\left[1 + \frac{e x^r}{d}\right] (a + b \operatorname{Log}[c x^n])^{p-1}}{x} dx$$

– Program code:

```
(* Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(x_*(d_+e_.*x_^r_.)),x_Symbol] :=
  (r*Log[x]-Log[1+(e*x^r)/d])*(a+b*Log[c*x^n])^p/(d*r) -
  b*n*p/d*Int[Log[x]*(a+b*Log[c*x^n])^(p-1)/x,x] +
  b*n*p/(d*r)*Int[Log[1+(e*x^r)/d]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0] *)
```

3: $\int \frac{(a + b \, Log[c \, x^n])^p}{x \, (d + e \, x^r)} \, dx$ when $p \in \mathbb{Z}^+$

Rule: Integration by parts

Basis: $\frac{1}{x \, (d+e \, x^r)} == -\frac{1}{d \, r} \partial_x \, Log\left[1 + \frac{d}{e \, x^r}\right]$

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \, Log[c \, x^n])^p}{x \, (d + e \, x^r)} \, dx \longrightarrow -\frac{Log\left[1 + \frac{d}{e \, x^r}\right] (a + b \, Log[c \, x^n])^p}{d \, r} + \frac{b \, n \, p}{d \, r} \int \frac{Log\left[1 + \frac{d}{e \, x^r}\right] (a + b \, Log[c \, x^n])^{p-1}}{x} \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./(x_*(d_+e_.*x_^r_.)),x_Symbol] :=
  -Log[1+d/(e*x^r)]*(a+b*Log[c*x^n])^p/(d*r) +
  b*n*p/(d*r)*Int[Log[1+d/(e*x^r)]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0]
```

2. $\int \frac{(d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

1: $\int \frac{(d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx$ when $p\in\mathbb{Z}^+\land q>0$

Rule: Algebraic expansion

Basis: $\frac{(d+e\,x)^q}{x} == \frac{d\,(d+e\,x)^{q-1}}{x} + e\,(d+e\,x)^{q-1}$

Rule: If $p\in\mathbb{Z}^+\land q>0$, then

$$\int \frac{(d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx \longrightarrow d\int \frac{(d+e\,x)^{q-1}\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx + e\int (d+e\,x)^{q-1}\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  d*Int[(d+e*x)^(q-1)*(a+b*Log[c*x^n])^p/x,x] +
  e*Int[(d+e*x)^(q-1)*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0] && GtQ[q,0] && IntegerQ[2*q]
```

2: $\int \frac{(d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx$ when $p\in\mathbb{Z}^+\land q<-1$

Rule: Algebraic expansion

Basis: $\frac{(d+e\,x)^q}{x}=\frac{(d+e\,x)^{q+1}}{d\,x}-\frac{e\,(d+e\,x)^q}{d}$

Rule: If $p\in\mathbb{Z}^+\land q<-1$, then

$$\int \frac{(d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx\longrightarrow \frac{1}{d}\int \frac{(d+e\,x)^{q+1}\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx-\frac{e}{d}\int (d+e\,x)^q\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  1/d*Int[(d+e*x)^(q+1)*(a+b*Log[c*x^n])^p/x,x] -
  e/d*Int[(d+e*x)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[p,0] && LtQ[q,-1] && IntegerQ[2*q]
```

3: $\int \frac{(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])}{x}\,dx$ when $q-\frac{1}{2}\in\mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\mathrm{Log}[c\,x^n]) == \frac{b\,n}{x}$

Rule: If $q-\frac{1}{2}\in\mathbb{Z}$, let $u\to\int\frac{(d+e\,x^r)^q}{x}\,dx$, then

$$\int \frac{(d+e\,x^r)^q\,(a+b\,\mathrm{Log}[c\,x^n])}{x}\,dx \longrightarrow u\,(a+b\,\mathrm{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])/x_,x_Symbol] :=
  With[{u=IntHide[(d+e*x^r)^q/x,x]},
  u*(a+b*Log[c*x^n]) - b*n*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,n,r},x] && IntegerQ[q-1/2]
```

4: $\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p \in \mathbb{Z}^+ \land q+1 \in \mathbb{Z}^-$

Rule: Algebraic expansion

Basis: $\frac{(d+e\,x^r)^q}{x} = \frac{(d+e\,x^r)^{q+1}}{d\,x} - \frac{e\,x^{r-1}\,(d+e\,x^r)^q}{d}$

Rule: If $p \in \mathbb{Z}^+ \land q+1 \in \mathbb{Z}^-$, then

$$\int \frac{(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx \longrightarrow \frac{1}{d}\int \frac{(d+e\,x^r)^{q+1}\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx - \frac{e}{d}\int x^{r-1}\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^r_.)^q_*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  1/d*Int[(d+e*x^r)^(q+1)*(a+b*Log[c*x^n])^p/x,x] -
  e/d*Int[x^(r-1)*(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,n,r},x] && IGtQ[p,0] && ILtQ[q,-1]
```

7: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $m\in\mathbb{Z}\,\land\,2\,q\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n]) == \frac{b\,n}{x}$

Note: If $m\in\mathbb{Z}\,\land\,q-\frac{1}{2}\in\mathbb{Z}$, then the terms of $\int x^m\,(d+e\,x)^q\,dx$ will be algebraic functions or constants times an inverse function.

Rule: If $m\in\mathbb{Z}\,\land\,2\,q\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$, let $u\to\int (f\,x)^m\,(d+e\,x^r)^q\,dx$, then

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])\,dx\;\longrightarrow\;u\,(a+b\,\text{Log}[c\,x^n])-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^r)^q,x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[SimplifyIntegrand[u/x,x],x] /;
 (EqQ[r,1] || EqQ[r,2]) && IntegerQ[m] && IntegerQ[q-1/2] || InverseFunctionFreeQ[u,x]] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && IntegerQ[2*q] && (IntegerQ[m] && IntegerQ[r] || IGtQ[q,0])
```

8: $\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx$ when $q\in\mathbb{Z}\wedge(q>0\vee m\in\mathbb{Z}\wedge r\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $q\in\mathbb{Z}\wedge(q>0\vee m\in\mathbb{Z}\wedge r\in\mathbb{Z})$, then

$$\int (f\,x)^m \left(d+e\,x^r\right)^q \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx \longrightarrow \int \left(a+b\,\text{Log}\left[c\,x^n\right]\right) \text{ExpandIntegrand}\left[(f\,x)^m \left(d+e\,x^r\right)^q,\,x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*x^n]),(f*x)^m*(d+e*x^r)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,m,n,q,r},x] && IntegerQ[q] && (GtQ[q,0] || IntegerQ[m] && IntegerQ[r])
```

9: $\int x^m\left(d+e\,x^r\right)^q\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^p\,dx$ when $q\in\mathbb{Z}\ \land\ \frac{r}{n}\in\mathbb{Z}\ \land\ \frac{m+1}{n}\in\mathbb{Z}\ \land\ \left(\frac{m+1}{n}>0\ \lor\ p\in\mathbb{Z}^+\right)$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F\left[x^n\right]==\frac{1}{n}\,\mathrm{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Rule: If $q\in\mathbb{Z}\ \land\ \frac{r}{n}\in\mathbb{Z}\ \land\ \frac{m+1}{n}\in\mathbb{Z}\ \land\ \left(\frac{m+1}{n}>0\ \lor\ p\in\mathbb{Z}^+\right)$, then

$$\int x^m\left(d+e\,x^r\right)^q\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^p\,dx\ \longrightarrow\ \frac{1}{n}\,\mathrm{Subst}\left[\int x^{\frac{m+1}{n}-1}\left(d+e\,x^{\frac{r}{n}}\right)^q\left(a+b\,\mathrm{Log}[c\,x]\right)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(d+e*x^(r/n))^q*(a+b*Log[c*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,e,m,n,p,q,r},x] && IntegerQ[q] && IntegerQ[r/n] && IntegerQ[Simplify[(m+1)/n]] && (GtQ[(m+1)/n,0] || IGtQ[p,0])
```

10: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $q\in\mathbb{Z}\land(q>0\lor p\in\mathbb{Z}^+\land m\in\mathbb{Z}\land r\in\mathbb{Z})$

– Derivation: Algebraic expansion

– Rule: If $q\in\mathbb{Z}\land(q>0\lor p\in\mathbb{Z}^+\land m\in\mathbb{Z}\land r\in\mathbb{Z})$, then

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx\to\int (a+b\,\text{Log}[c\,x^n])^p\,\text{ExpandIntegrand}[(f\,x)^m\,(d+e\,x^r)^q,\,x]\,dx$$

– Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*x^n])^p,(f*x)^m*(d+e*x^r)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && IntegerQ[q] && (GtQ[q,0] || IGtQ[p,0] && IntegerQ[m] && IntegerQ[r])
```

U: $\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

– Rule:

$$\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx\to\int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

– Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^r)^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x]
```

N: $\int (f\,x)^m\,u^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $u == d+e\,x^r$

Derivation: Algebraic normalization

–

Rule: If $u == d+e\,x^r$, then

$$\int (f\,x)^m\,u^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx \to \int (f\,x)^m\,(d+e\,x^r)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

–

Program code:

```
Int[(f_.*x_)^m_.*u_^q_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[(f*x)^m*ExpandToSum[u,x]^q*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,f,m,n,p,q},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

5. $\int \text{AF}[x]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

1: $\int \text{Poly}[x]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

Derivation: Algebraic expansion

–

Rule:

$$\int \text{Poly}[x]\,(a+b\,\text{Log}[c\,x^n])^p\,dx \to \int \text{ExpandIntegrand}\left[\text{Poly}[x]\,(a+b\,\text{Log}[c\,x^n])^p,\,x\right]dx$$

–

Program code:

```
Int[Polyx_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Polyx*(a+b*Log[c*x^n])^p,x],x] /;
FreeQ[{a,b,c,n,p},x] && PolynomialQ[Polyx,x]
```

2: $\int RF[x]\,\left(a+b\,Log\left[c\,x^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int RF[x]\,\left(a+b\,Log\left[c\,x^n\right]\right)^p\,dx\;\longrightarrow\;\int\left(a+b\,Log\left[c\,x^n\right]\right)^p\,\text{ExpandIntegrand}[RF[x],\,x]\,dx$$

Program code:

```
Int[RFx_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*x^n])^p,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,n},x] && RationalFunctionQ[RFx,x] && IGtQ[p,0]
```

```
Int[RFx_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[RFx*(a+b*Log[c*x^n])^p,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,n},x] && RationalFunctionQ[RFx,x] && IGtQ[p,0]
```

U: $\int AF[x]\,(a+b\,Log[c\,x^n])^p\,dx$

Rule:

$$\int AF[x]\,(a+b\,Log[c\,x^n])^p\,dx \longrightarrow \int AF[x]\,(a+b\,Log[c\,x^n])^p\,dx$$

Program code:

```
Int[AFx_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[AFx*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,n,p},x] && AlgebraicFunctionQ[AFx,x,True]
```

6. $\int (a+b\,Log[c\,x^n])^p\,(d+e\,Log[f\,x^r])^q\,dx$

1: $\int (a+b\,Log[c\,x^n])^p\,(d+e\,Log[c\,x^n])^q\,dx$ when $p\in\mathbb{Z}\land q\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}\land q\in\mathbb{Z}$, then

$$\int (a+b\,Log[c\,x^n])^p\,(d+e\,Log[c\,x^n])^q\,dx \longrightarrow \int ExpandIntegrand\left[(a+b\,Log[c\,x^n])^p\,(d+e\,Log[c\,x^n])^q,\,x\right]dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_+e_.*Log[c_.*x_^n_.])^q_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*x^n])^p*(d+e*Log[c*x^n])^q,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && IntegerQ[p] && IntegerQ[q]
```

2: $\int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r]) \, dx$

Derivation: Integration by parts

Rule: Let $u \to \int (a + b \operatorname{Log}[c x^n])^p \, dx$, then

$$\int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r]) \, dx \longrightarrow u (d + e \operatorname{Log}[f x^r]) - e r \int \frac{u}{x} \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[f_.*x_^r_.]),x_Symbol] :=
  With[{u=IntHide[(a+b*Log[c*x^n])^p,x]},
  Dist[d+e*Log[f*x^r],u,x] - e*r*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,f,n,p,r},x]
```

3: $\int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^q \, dx$ when $p \in \mathbb{Z}^+ \wedge q \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $p \in \mathbb{Z}^+ \wedge q \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^q \, dx \longrightarrow$$

$$x (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^q - e q r \int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^{q-1} \, dx - b n p \int (a + b \operatorname{Log}[c x^n])^{p-1} (d + e \operatorname{Log}[f x^r])^q \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[f_.*x_^r_.])^q_.,x_Symbol] :=
  x*(a+b*Log[c*x^n])^p*(d+e*Log[f*x^r])^q -
  e*q*r*Int[(a+b*Log[c*x^n])^p*(d+e*Log[f*x^r])^(q-1),x] -
  b*n*p*Int[(a+b*Log[c*x^n])^(p-1)*(d+e*Log[f*x^r])^q,x] /;
FreeQ[{a,b,c,d,e,f,n,r},x] && IGtQ[p,0] && IGtQ[q,0]
```

U: $\int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^q \, dx$

Rule:

$$\int (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^q \, dx \longrightarrow \int (g x)^m (a + b \operatorname{Log}[c x^n])^p (d + e \operatorname{Log}[f x^r])^q \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[f_.*x_^r_.])^q_.,x_Symbol] :=
  Unintegrable[(a+b*Log[c*x^n])^p*(d+e*Log[f*x^r])^q,x] /;
FreeQ[{a,b,c,d,e,f,n,p,q,r},x]
```

S: $\int (a + b \text{Log}[v])^p (c + d \text{Log}[v])^q \, dx$ when $v == g + h x \land g \neq 0$

Derivation: Integration by substitution

Rule: If $v == g + h x \land g \neq 0$, then

$$\int (a + b \text{Log}[v])^p (c + d \text{Log}[v])^q \, dx \longrightarrow \frac{1}{h} \text{Subst}\left[\int (a + b \text{Log}[x])^p (c + d \text{Log}[x])^q \, dx, x, g + h x\right]$$

Program code:

```
Int[(a_.+b_.*Log[v_])^p_.*(c_.+d_.*Log[v_])^q_.,x_Symbol] :=
  1/Coeff[v,x,1]*Subst[Int[(a+b*Log[x])^p*(c+d*Log[x])^q,x],x,v] /;
FreeQ[{a,b,c,d,p,q},x] && LinearQ[v,x] && NeQ[Coeff[v,x,0],0]
```

7. $\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q\,dx$

1: $\int \frac{(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[c\,x^n])^q}{x}\,dx$

Derivation: Integration by substitution

Basis: $\frac{F[\text{Log}[c\,x^n]]}{x} == \frac{1}{n}\,\text{Subst}[F[x],\,x,\,\text{Log}[c\,x^n]]\,\partial_x\,\text{Log}[c\,x^n]$

Rule:

$$\int \frac{(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[c\,x^n])^q}{x}\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int (a+b\,x)^p\,(d+e\,x)^q\,dx,\,x,\,\text{Log}[c\,x^n]\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[c_.*x_^n_.])^q_./x_,x_Symbol] :=
  1/n*Subst[Int[(a+b*x)^p*(d+e*x)^q,x],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,e,n,p,q},x]
```

2: $\int (g\,x)^m\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,\left(d+e\,\text{Log}\left[f\,x^r\right]\right)\,dx$

Derivation: Integration by parts

Rule: Let $u\to\int (g\,x)^m\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$, then

$$\int (g\,x)^m\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,\left(d+e\,\text{Log}\left[f\,x^r\right]\right)\,dx\;\longrightarrow\;u\,\left(d+e\,\text{Log}\left[f\,x^r\right]\right)-e\,r\int\frac{u}{x}\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[f_.*x_^r_.]),x_Symbol] :=
  With[{u=IntHide[(g*x)^m*(a+b*Log[c*x^n])^p,x]},
  Dist[(d+e*Log[f*x^r]),u,x] - e*r*Int[SimplifyIntegrand[u/x,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,r},x] && Not[EqQ[p,1] && EqQ[a,0] && NeQ[d,0]]
```

3: $\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q\,dx$ when $p\in\mathbb{Z}^+\land q\in\mathbb{Z}^+\land m\neq-1$

Derivation: Integration by parts

Rule: If $p\in\mathbb{Z}^+\land q\in\mathbb{Z}^+\land m\neq-1$, then

$$\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q\,dx\;\longrightarrow$$

$$\frac{(g\,x)^{m+1}\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q}{g\,(m+1)}-\frac{e\,q\,r}{m+1}\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^{q-1}\,dx-\frac{b\,n\,p}{m+1}\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^{p-1}\,(d+e\,\text{Log}[f\,x^r])^q\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[f_.*x_^r_.])^q_.,x_Symbol] :=
  (g*x)^(m+1)*(a+b*Log[c*x^n])^p*(d+e*Log[f*x^r])^q/(g*(m+1)) -
  e*q*r/(m+1)*Int[(g*x)^m*(a+b*Log[c*x^n])^p*(d+e*Log[f*x^r])^(q-1),x] -
  b*n*p/(m+1)*Int[(g*x)^m*(a+b*Log[c*x^n])^(p-1)*(d+e*Log[f*x^r])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,r},x] && IGtQ[p,0] && IGtQ[q,0] && NeQ[m,-1]
```

U: $\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q\,dx$

Rule:

$$\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q\,dx\;\longrightarrow\;\int (g\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,(d+e\,\text{Log}[f\,x^r])^q\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.*(d_.+e_.*Log[f_.*x_^r_.])^q_.,x_Symbol] :=
  Unintegrable[(g*x)^m*(a+b*Log[c*x^n])^p*(d+e*Log[f*x^r])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r},x]
```

S: $\int u^m (a + b \operatorname{Log}[v])^p (c + d \operatorname{Log}[v])^q \, dx$ when $u == e + f x \land v == g + h x \land f g - e h == 0 \land g \neq 0$

Derivation: Integration by substitution

Rule: If $u == e + f x \land v == g + h x \land f g - e h == 0 \land g \neq 0$, then

$$\int u^m (a + b \operatorname{Log}[v])^p (c + d \operatorname{Log}[v])^q \, dx \longrightarrow \frac{1}{h} \operatorname{Subst}\left[\int \left(\frac{f x}{h}\right)^m (a + b \operatorname{Log}[x])^p (c + d \operatorname{Log}[x])^q \, dx, x, g + h x\right]$$

Program code:

```
Int[u_^m_.*(a_.+b_.*Log[v_])^p_.*(c_.+d_.*Log[v_])^q_.,x_Symbol] :=
  With[{e=Coeff[u,x,0],f=Coeff[u,x,1],g=Coeff[v,x,0],h=Coeff[v,x,1]},
  1/h*Subst[Int[(f*x/h)^m*(a+b*Log[x])^p*(c+d*Log[x])^q,x],x,v] /;
 EqQ[f*g-e*h,0] && NeQ[g,0]] /;
FreeQ[{a,b,c,d,m,p,q},x] && LinearQ[{u,v},x]
```

8. $\int \text{Log}[d\,(e+f\,x^m)^r]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

1: $\int \text{Log}[d\,(e+f\,x^m)^r]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+ \wedge m\in\mathbb{R} \wedge \left(p==1 \vee \frac{1}{m}\in\mathbb{Z} \vee r==1 \wedge m==1 \wedge d\,e==1\right)$

Derivation: Integration by parts

Note: If $m\in\mathbb{R}$, then $\frac{\int \text{Log}[d\,(e+f\,x^m)^r]\,dx}{x}$ is integrable.

Rule: If $p\in\mathbb{Z}^+ \wedge m\in\mathbb{R} \wedge \left(p==1 \vee \frac{1}{m}\in\mathbb{Z} \vee r==1 \wedge m==1 \wedge d\,e==1\right)$, let $u\to\int \text{Log}[d\,(e+f\,x^m)^r]\,dx$, then

$$\int \text{Log}[d\,(e+f\,x^m)^r]\,(a+b\,\text{Log}[c\,x^n])^p\,dx \longrightarrow u\,(a+b\,\text{Log}[c\,x^n])^p - b\,n\,p\int \frac{u\,(a+b\,\text{Log}[c\,x^n])^{p-1}}{x}\,dx$$

Program code:

```
Int[Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=IntHide[Log[d*(e+f*x^m)^r],x]},
  Dist[(a+b*Log[c*x^n])^p,u,x] - b*n*p*Int[Dist[(a+b*Log[c*x^n])^(p-1)/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,r,m,n},x] && IGtQ[p,0] && RationalQ[m] && (EqQ[p,1] || FractionQ[m] && IntegerQ[1/m] || EqQ[r,1] && EqQ[m,1] && EqQ[d*e,
```

2: $\int \text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$ when $p\in\mathbb{Z}^+\land m\in\mathbb{Z}$

Derivation: Integration by parts

Rule: If $p\in\mathbb{Z}^+\land m\in\mathbb{Z}$, let $u\to\int\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$, then

$$\int \text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx \longrightarrow u\,\text{Log}\left[d\left(e+f x^m\right)^r\right]-f\,m\,r\int\frac{u\,x^{m-1}}{e+f x^m}\,dx$$

Program code:

```
Int[Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=IntHide[(a+b*Log[c*x^n])^p,x]},
  Dist[Log[d*(e+f*x^m)^r],u,x] - f*m*r*Int[Dist[x^(m-1)/(e+f*x^m),u,x],x]] /;
FreeQ[{a,b,c,d,e,f,r,m,n},x] && IGtQ[p,0] && IntegerQ[m]
```

U: $\int \text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$

Rule:

$$\int \text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx \longrightarrow \int \text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx$$

Program code:

```
Int[Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[Log[d*(e+f*x^m)^r]*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,r,m,n,p},x]
```

N: $\int \text{Log}[d\,u^r]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $u == e+f\,x^m$

Derivation: Algebraic normalization

Rule: If $u == e+f\,x^m$, then

$$\int (g\,x)^q\,\text{Log}[d\,u^r]\,(a+b\,\text{Log}[c\,x^n])^p\,dx \longrightarrow \int (g\,x)^q\,\text{Log}[d\,(e+f\,x^m)^r]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[Log[d_.*u_^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[Log[d*ExpandToSum[u,x]^r]*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,r,n,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

9. $\int (g\,x)^q\,\text{Log}\left[d\,\left(e+f\,x^m\right)^r\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$

1. $\int \frac{\text{Log}\left[d\,\left(e+f\,x^m\right)^r\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

1: $\int \frac{\text{Log}\left[d\,\left(e+f\,x^m\right)\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx$ when $p\in\mathbb{Z}^+\land\,d\,e==1$

Derivation: Integration by parts

Basis: If $d\,e==1$, then $\frac{\text{Log}[d\,(e+f\,x^m)]}{x}==-\partial_x\frac{\text{PolyLog}[2,-d\,f\,x^m]}{m}$

Rule: If $p\in\mathbb{Z}^+\land\,d\,e==1$, then

$$\int \frac{\text{Log}\left[d\,\left(e+f\,x^m\right)\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx\;\longrightarrow\;-\frac{\text{PolyLog}\left[2,\,-d\,f\,x^m\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{m}+\frac{b\,n\,p}{m}\int \frac{\text{PolyLog}\left[2,\,-d\,f\,x^m\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p-1}}{x}\,dx$$

Program code:

```
Int[Log[d_.*(e_+f_.*x_^m_.)]*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  -PolyLog[2,-d*f*x^m]*(a+b*Log[c*x^n])^p/m +
  b*n*p/m*Int[PolyLog[2,-d*f*x^m]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0] && EqQ[d*e,1]
```

2: $\int \frac{\text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx$ when $p \in \mathbb{Z}^+ \land d\,e \neq 1$

Derivation: Integration by parts

Basis: $\frac{\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x} == \partial_x \frac{\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p+1}}{b\,n\,(p+1)}$

Basis: $\partial_x \text{Log}\left[d\left(e+f x^m\right)^r\right] == \frac{f\,m\,r\,x^{m-1}}{e+f x^m}$

Rule: If $p \in \mathbb{Z}^+ \land d\,e \neq 1$, then

$$\int \frac{\text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p}{x}\,dx \longrightarrow \frac{\text{Log}\left[d\left(e+f x^m\right)^r\right]\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p+1}}{b\,n\,(p+1)} - \frac{f\,m\,r}{b\,n\,(p+1)}\int \frac{x^{m-1}\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p+1}}{e+f x^m}\,dx$$

Program code:

```
Int[Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  Log[d*(e+f*x^m)^r]*(a+b*Log[c*x^n])^(p+1)/(b*n*(p+1)) -
  f*m*r/(b*n*(p+1))*Int[x^(m-1)*(a+b*Log[c*x^n])^(p+1)/(e+f*x^m),x] /;
FreeQ[{a,b,c,d,e,f,r,m,n},x] && IGtQ[p,0] && NeQ[d*e,1]
```

2: $\int (g\,x)^q \operatorname{Log}\left[d\,\left(e+f\,x^m\right)^r\right]\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\,dx$ when $\left(\frac{q+1}{m}\in\mathbb{Z}\ \lor\ (m\mid q)\in\mathbb{R}\right)\land q\neq -1$

Derivation: Integration by parts

Note: If $\frac{q+1}{m}\in\mathbb{Z}\ \lor\ (m\mid q)\in\mathbb{R}$, then $\frac{\int (g\,x)^q \operatorname{Log}\left[d\,\left(e+f\,x^m\right)^r\right]\,dx}{x}$ is integrable.

Rule: If $\left(\frac{q+1}{m}\in\mathbb{Z}\ \lor\ (m\mid q)\in\mathbb{R}\right)\land q\neq -1$, let $u\to\int (g\,x)^q \operatorname{Log}\left[d\,\left(e+f\,x^m\right)^r\right]\,dx$, then

$$\int (g\,x)^q \operatorname{Log}\left[d\,\left(e+f\,x^m\right)^r\right]\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\,dx\ \longrightarrow\ u\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)-b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[(g*x)^q*Log[d*(e+f*x^m)^r],x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,r,m,n,q},x] && (IntegerQ[(q+1)/m] || RationalQ[m] && RationalQ[q]) && NeQ[q,-1]
```

3: $\int (g\,x)^q \,\text{Log}\left[d\,\left(e+f\,x^m\right)\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}^+\wedge m\in\mathbb{R}\wedge q\in\mathbb{R}\wedge q\neq -1\wedge\left(p==1\vee\frac{q+1}{m}\in\mathbb{Z}\vee\left(q\in\mathbb{Z}^+\wedge\frac{q+1}{m}\in\mathbb{Z}\wedge d\,e==1\right)\right)$

Derivation: Integration by parts

– Rule: If $p\in\mathbb{Z}^+\wedge m\in\mathbb{R}\wedge q\in\mathbb{R}\wedge q\neq -1\wedge\left(p==1\vee\frac{q+1}{m}\in\mathbb{Z}\vee\left(q\in\mathbb{Z}^+\wedge\frac{q+1}{m}\in\mathbb{Z}\wedge d\,e==1\right)\right)$, let $u\to\int (g\,x)^q\,\text{Log}\left[d\,\left(e+f\,x^m\right)\right]\,dx$, then

$$\int (g\,x)^q\,\text{Log}\left[d\,\left(e+f\,x^m\right)\right]\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p\,dx\;\longrightarrow\;u\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p-b\,n\,p\int\frac{u\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p-1}}{x}\,dx$$

– Program code:

```
Int[(g_.*x_)^q_.*Log[d_.*(e_+f_.*x_^m_.)]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=IntHide[(g*x)^q*Log[d*(e+f*x^m)],x]},
  Dist[(a+b*Log[c*x^n])^p,u,x] - b*n*p*Int[Dist[(a+b*Log[c*x^n])^(p-1)/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && IGtQ[p,0] && RationalQ[m] && RationalQ[q] && NeQ[q,-1] &&
  (EqQ[p,1] || FractionQ[m] && IntegerQ[(q+1)/m] || IGtQ[q,0] && IntegerQ[(q+1)/m] && EqQ[d*e,1])
```

4: $\int (g\,x)^q \,\text{Log}\left[d\,(e+f\,x^m)^r\right]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+\land m\in\mathbb{R}\,\land\,q\in\mathbb{R}$

Derivation: Integration by parts

Rule: If $p\in\mathbb{Z}^+\land\,m\in\mathbb{R}\,\land\,q\in\mathbb{R}$, let $u\to\int (g\,x)^q\,(a+b\,\text{Log}[c\,x^n])^p\,dx$, then

$$\int (g\,x)^q\,\text{Log}\left[d\,(e+f\,x^m)^r\right]\,(a+b\,\text{Log}[c\,x^n])^p\,dx\;\longrightarrow\;u\,\text{Log}\left[d\,(e+f\,x^m)^r\right]-f\,m\,r\int\frac{u\,x^{m-1}}{e+f\,x^m}\,dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  With[{u=IntHide[(g*x)^q*(a+b*Log[c*x^n])^p,x]},
  Dist[Log[d*(e+f*x^m)^r],u,x] - f*m*r*Int[Dist[x^(m-1)/(e+f*x^m),u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,r,m,n,q},x] && IGtQ[p,0] && RationalQ[m] && RationalQ[q]
```

U: $\int (g\,x)^q\,\text{Log}\left[d\,(e+f\,x^m)^r\right]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

Rule:

$$\int (g\,x)^q\,\text{Log}\left[d\,(e+f\,x^m)^r\right]\,(a+b\,\text{Log}[c\,x^n])^p\,dx\;\longrightarrow\;\int (g\,x)^q\,\text{Log}\left[d\,(e+f\,x^m)^r\right]\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[d_.*(e_+f_.*x_^m_.)^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(g*x)^q*Log[d*(e+f*x^m)^r]*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,r,m,n,p,q},x]
```

N: $\int (g\,x)^q \operatorname{Log}\left[d\,u^r\right] \left(a+b \operatorname{Log}\left[c\,x^n\right]\right)^p dx$ when $u == e+f\,x^m$

– Derivation: Algebraic normalization

– Rule: If $u == e+f\,x^m$, then

$$\int (g\,x)^q \operatorname{Log}\left[d\,u^r\right] \left(a+b \operatorname{Log}\left[c\,x^n\right]\right)^p dx \longrightarrow \int (g\,x)^q \operatorname{Log}\left[d\left(e+f\,x^m\right)^r\right] \left(a+b \operatorname{Log}\left[c\,x^n\right]\right)^p dx$$

– Program code:

```
Int[(g_.*x_)^q_.*Log[d_.*u_^r_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Int[(g*x)^q*Log[d*ExpandToSum[u,x]^r]*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,g,r,n,p,q},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

10. $\int \text{PolyLog}[k, e\, x^q]\, (a + b\, \text{Log}[c\, x^n])^p\, dx$

1: $\int \text{PolyLog}[k, e\, x^q]\, (a + b\, \text{Log}[c\, x^n])\, dx$ when $k \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $(a + b\, \text{Log}[c\, x^n]) == \partial_x (-b\, n\, x + x\, (a + b\, \text{Log}[c\, x^n]))$

Basis: $\partial_x \text{PolyLog}[k, e\, x^q] == \frac{q\, \text{PolyLog}[k-1, e\, x^q]}{x}$

Rule: If $k \in \mathbb{Z}^+$, then

$$\int \text{PolyLog}[k, e\, x^q]\, (a + b\, \text{Log}[c\, x^n])\, dx \longrightarrow$$

$$-b\, n\, x\, \text{PolyLog}[k, e\, x^q] + x\, \text{PolyLog}[k, e\, x^q]\, (a + b\, \text{Log}[c\, x^n]) + b\, n\, q \int \text{PolyLog}[k-1, e\, x^q]\, dx - q \int \text{PolyLog}[k-1, e\, x^q]\, (a + b\, \text{Log}[c\, x^n])\, dx$$

Program code:

```
Int[PolyLog[k_,e_.*x_^q_.]*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  -b*n*x*PolyLog[k,e*x^q] + x*PolyLog[k,e*x^q]*(a+b*Log[c*x^n]) +
  b*n*q*Int[PolyLog[k-1,e*x^q],x] - q*Int[PolyLog[k-1,e*x^q]*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,e,n,q},x] && IGtQ[k,0]
```

U: $\int \text{PolyLog}[k, e\,x^q]\,(a + b\,\text{Log}[c\,x^n])^p\,dx$

Rule:

$$\int \text{PolyLog}[k, e\,x^q]\,(a + b\,\text{Log}[c\,x^n])^p\,dx \longrightarrow \int \text{PolyLog}[k, e\,x^q]\,(a + b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[PolyLog[k_,e_.*x_^q_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[PolyLog[k,e*x^q]*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,e,n,p,q},x]
```

11. $\int (d\,x)^m \,\text{PolyLog}[k,\, e\,x^q]\, (a + b\,\text{Log}[c\,x^n])^p \, dx$

1. $\int \frac{\text{PolyLog}[k,\, e\,x^q]\, (a + b\,\text{Log}[c\,x^n])^p}{x} \, dx$

1: $\int \frac{\text{PolyLog}[k,\, e\,x^q]\, (a + b\,\text{Log}[c\,x^n])^p}{x} \, dx$ when $p > 0$

Derivation: Integration by parts

– Basis: $\frac{\text{PolyLog}[k,e\,x^q]}{x} == \partial_x \frac{\text{PolyLog}[k+1,e\,x^q]}{q}$

– Rule: If $p > 0$, then

$$\int \frac{\text{PolyLog}[k,\, e\,x^q]\, (a + b\,\text{Log}[c\,x^n])^p}{x} \, dx \longrightarrow$$
$$\frac{\text{PolyLog}[k+1,\, e\,x^q]\, (a + b\,\text{Log}[c\,x^n])^p}{q} - \frac{b\,n\,p}{q} \int \frac{\text{PolyLog}[k+1,\, e\,x^q]\, (a + b\,\text{Log}[c\,x^n])^{p-1}}{x} \, dx$$

– Program code:

```
Int[PolyLog[k_,e_.*x_^q_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  PolyLog[k+1,e*x^q]*(a+b*Log[c*x^n])^p/q - b*n*p/q*Int[PolyLog[k+1,e*x^q]*(a+b*Log[c*x^n])^(p-1)/x,x] /;
FreeQ[{a,b,c,e,k,n,q},x] && GtQ[p,0]
```

2: $\int \frac{\text{PolyLog}[k, e\,x^q]\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx$ when $p < -1$

Derivation: Integration by parts

- Basis: $\frac{(a+b\,\text{Log}[c\,x^n])^p}{x} == \partial_x \frac{(a+b\,\text{Log}[c\,x^n])^{p+1}}{b\,n\,(p+1)}$

- Basis: $\partial_x\,\text{PolyLog}[k, e\,x^q] == \frac{q\,\text{PolyLog}[k-1, e\,x^q]}{x}$

- Rule: If $p < -1$, then

$$\int \frac{\text{PolyLog}[k, e\,x^q]\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx \longrightarrow$$
$$\frac{\text{PolyLog}[k, e\,x^q]\,(a+b\,\text{Log}[c\,x^n])^{p+1}}{b\,n\,(p+1)} - \frac{q}{b\,n\,(p+1)}\int \frac{\text{PolyLog}[k-1, e\,x^q]\,(a+b\,\text{Log}[c\,x^n])^{p+1}}{x}\,dx$$

Program code:

```
Int[PolyLog[k_,e_.*x_^q_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  PolyLog[k,e*x^q]*(a+b*Log[c*x^n])^(p+1)/(b*n*(p+1)) - q/(b*n*(p+1))*Int[PolyLog[k-1,e*x^q]*(a+b*Log[c*x^n])^(p+1)/x,x] /;
FreeQ[{a,b,c,e,k,n,q},x] && LtQ[p,-1]
```

2: $\int (d\,x)^m\,\text{PolyLog}[k, e\,x^q]\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $k \in \mathbb{Z}^+$

Derivation: Integration by parts

- Basis: $(d\,x)^m\,(a+b\,\text{Log}[c\,x^n]) == \partial_x \left(-\frac{b\,n\,(d\,x)^{m+1}}{d\,(m+1)^2} + \frac{(d\,x)^{m+1}\,(a+b\,\text{Log}[c\,x^n])}{d\,(m+1)}\right)$

- Basis: $\partial_x\,\text{PolyLog}[k, e\,x^q] == \frac{q\,\text{PolyLog}[k-1, e\,x^q]}{x}$

- Rule: If $k \in \mathbb{Z}^+$, then

$$\int (d\,x)^m\,\text{PolyLog}[k,\,e\,x^q]\,\left(a+b\,\text{Log}[c\,x^n]\right)\,dx\,\longrightarrow$$

$$-\frac{b\,n\,(d\,x)^{m+1}\,\text{PolyLog}[k,\,e\,x^q]}{d\,(m+1)^2}+\frac{(d\,x)^{m+1}\,\text{PolyLog}[k,\,e\,x^q]\,\left(a+b\,\text{Log}[c\,x^n]\right)}{d\,(m+1)}+$$
$$\frac{b\,n\,q}{(m+1)^2}\int (d\,x)^m\,\text{PolyLog}[k-1,\,e\,x^q]\,dx-\frac{q}{(m+1)}\int (d\,x)^m\,\text{PolyLog}[k-1,\,e\,x^q]\,\left(a+b\,\text{Log}[c\,x^n]\right)\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*PolyLog[k_,e_.*x_^q_.]*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  -b*n*(d*x)^(m+1)*PolyLog[k,e*x^q]/(d*(m+1)^2) +
   (d*x)^(m+1)*PolyLog[k,e*x^q]*(a+b*Log[c*x^n])/(d*(m+1)) +
  b*n*q/(m+1)^2*Int[(d*x)^m*PolyLog[k-1,e*x^q],x] -
  q/(m+1)*Int[(d*x)^m*PolyLog[k-1,e*x^q]*(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,m,n,q},x] && IGtQ[k,0]
```

U: $\int (d\,x)^m\,\text{PolyLog}[k,\,e\,x^q]\,\left(a+b\,\text{Log}[c\,x^n]\right)^p\,dx$

Rule:

$$\int (d\,x)^m\,\text{PolyLog}[k,\,e\,x^q]\,\left(a+b\,\text{Log}[c\,x^n]\right)^p\,dx\,\longrightarrow\,\int (d\,x)^m\,\text{PolyLog}[k,\,e\,x^q]\,\left(a+b\,\text{Log}[c\,x^n]\right)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*PolyLog[k_,e_.*x_^q_.]*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(d*x)^m*PolyLog[k,e*x^q]*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x]
```

12. $\int P_x F[d (e + f x)]^m (a + b \text{Log}[c x^n]) \, dx$

1: $\int P_x F[d (e + f x)]^m (a + b \text{Log}[c x^n]) \, dx$ when $m \in \mathbb{Z}^+ \land F \in \{\text{ArcSin}, \text{ArcCos}, \text{ArcSinh}, \text{ArcCosh}\}$

Derivation: Integration by parts

Basis: $\partial_x (a + b \text{Log}[c x^n]) == \frac{b n}{x}$

Note: If $m \in \mathbb{Z}^+ \land F \in \{\text{ArcSin}, \text{ArcCos}, \text{ArcSinh}, \text{ArcCosh}\}$, the terms of the antiderivative of $\frac{\int P_x F[d (e+f x)]^m dx}{x}$ will be integrable.

Rule: If $m \in \mathbb{Z}^+ \land F \in \{\text{ArcSin}, \text{ArcCos}, \text{ArcSinh}, \text{ArcCosh}\}$, let $u \to \int P_x F[d (e + f x)]^m dx$, then

$$\int P_x F[d (e + f x)]^m (a + b \text{Log}[c x^n]) \, dx \longrightarrow u (a + b \text{Log}[c x^n]) - b n \int \frac{u}{x} \, dx$$

Program code:

```
Int[Px_.*F_[d_.*(e_.+f_.*x_)]^m_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[Px*F[d*(e+f*x)]^m,x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && PolynomialQ[Px,x] && IGtQ[m,0] && MemberQ[{ArcSin, ArcCos, ArcSinh, ArcCosh},F]
```

2: $\int P_x\, F[d\,(e+f\,x)]\,(a+b\,\text{Log}[c\,x^n])\,dx$ when $F \in \{\text{ArcTan}, \text{ArcCot}, \text{ArcTanh}, \text{ArcCoth}\}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,x^n]) == \frac{b\,n}{x}$

Note: If $F \in \{\text{ArcTan}, \text{ArcCot}, \text{ArcTanh}, \text{ArcCoth}\}$, the terms of the antiderivative of $\frac{\int P_x\,F[d\,(e+f\,x)]\,dx}{x}$ will be integrable.

Rule: If $F \in \{\text{ArcTan}, \text{ArcCot}, \text{ArcTanh}, \text{ArcCoth}\}$, let $u \to \int P_x\,F[d\,(e+f\,x)]\,dx$, then

$$\int P_x\,F[d\,(e+f\,x)]\,(a+b\,\text{Log}[c\,x^n])\,dx \longrightarrow u\,(a+b\,\text{Log}[c\,x^n]) - b\,n\int\frac{u}{x}\,dx$$

Program code:

```
Int[Px_.*F_[d_.*(e_.+f_.*x_)]*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  With[{u=IntHide[Px*F[d*(e+f*x)],x]},
  Dist[(a+b*Log[c*x^n]),u,x] - b*n*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && PolynomialQ[Px,x] && MemberQ[{ArcTan, ArcCot, ArcTanh, ArcCoth},F]
```